
ABSTRACT 
We provide an IDL implementation of Fourier deconvolution and apply it to data from a 
WFC3 H1RG near-infrared array detector that exhibits inter-pixel capacitance (IPC). 
The deconvolution removes the most obvious deleterious effect of IPC: the cross-like 
pattern (i.e. the arithematic symbol for addition, “+”) of charge around pixels with large 
dark current, a.k.a. “hot” pixels. We also exhibit a case in which our physical 
interpretation of a detector defect changed qualitatively before and after the 
deconvolution: in a region of multiple “scratches” on the detector, before deconvolution 
we thought each “scratch” was spatially resolved by the 18-micron pixels, but after 
deconvolution, it appears that the width of each “scratch” is much smaller than a pixel. 
We briefly discuss a number of potential benefits of deconvolution of data from arrays 
with IPC: the preliminary analysis in this report may motivate and expedite additional 
simulations and observations upon which a more definitive and quantitative analysis will 
be based. In practice, deconvolution of IPC can be performed after images have been 
calibrated with the nominal (i.e. existing) pipeline process, although we discuss how in 
principle and potentially for optimum effect, the deconvolution of IPC should be 
performed on all data (science images, flat-fields, etc) at the outset, after reading out the 
arrays and prior to calibrating the science images with darks, flat-fields, cosmic-ray 
rejection, etc. A subsequent report will quantify the benefits of IPC-deconvolution prior 
to, instead of after, pixel-by-pixel calibration; this report only outlines the potential 
benefits qualitatively. 

Introduction 
Interpixel capacitance (IPC) is a deterministic cross-talk of charge in array detectors 

such as those commonly used in the near-infrared, e.g. WFC3 IR and JWST’s FGS, 
NIRSpec, and NIRCam.1 IPC spreads charge from pixels in which it was collected into 
adjacent pixels prior to readout. IPC has been modeled as a convolution with a small 
kernel (Finger et al. 2005, Moore, Ninkov, & Forrest 2006, and McCullough et al. 2008). 
The smoothing effect of IPC can make the Poisson noise of extended sources appear 
lower than it actually is, which can be misinterpretted as the array collecting more 
photons than it actually does. This paper relies on the deterministic nature of the analog 
charge re-apportionment inherent to IPC and on the convolution model for IPC. Those 
assumptions may break down, i.e. the detector’s physical IPC effect may not precisely act 
as a convolution, or the convolution kernel may depend on location on the array or on the 
charge on nearby pixels. Some future algorithm may account for those “second-order” 
effects, much as they do for an imaging system with a spatially-dependent PSF. Although 
knowledge of the parameterization of the IPC kernel may be inaccurate, correcting for the 
IPC effect by a simple, global deconvolution as described in this paper is likely to do 
more good than harm. Also, one can use the deconvolution in an iterative feedback cycle 
to better parameterize the IPC kernel. 

Deconvolution Algorithm 
The deconvolution algorithm is very simple.2 In the nominal model of IPC described  in 
the references in the Introduction, the observed array A’ is the convolution (* symbol) of 
the true image array A with the IPC kernel k: 
 

(1) k =  

! 

0 " 0

" 1# 4" "

0 " 0

 

 
(2) A’ = A * k. 

 
The kernel k is a 3-by-3 matrix, parameterized by alpha and optionally beta (in this work 
beta=alpha) that quantify the capacitive coupling of the center pixel’s charge to each of 
its adjacent neighbors along columns (alpha) or rows (beta). The four corner elements of 
k are zero, and the sum of elements of k is 1.  

                                                             
1 Abbreviations: Wide Field Camera 3 Infra-Red channel; James Webb Space Telescope; 
Fine Guidance Sensor; Near-Infra-Red Spectrograph; Nar-Infra-Red Camera. 
2 In this report, the images, true and observed, refer to the irradiance on the detector in its 
(X,Y) coordinates; any transformations such as distortion correction or deconvolution of 
the telescope’s PSF are not part of this analysis. 

Equation 2 can be Fourier transformed (F in Equation 3), then rearranged and inverse 
Fourier transformed (F-1) to recover the true array (Equation 4):  

(3) F(A’) = F(A * k) = F(A) F(k) 
(4) A = F-1 ( F(A’) / F(k) ) = Dk(A’). 

For convenience in later discussions, we have defined the deconvolution operator Dk in 
the latter equation. Here and later, we state without attribution properties of Fourier 
transforms and convolution (see e.g. Bracewell 2000).  Because k is very compact, its 
Fourier transform F(k) is very broad, indeed so broad that no where does it contain a 
zero, which enables the direct Fourier deconvolution of Equation 4 even though F(k) 
appears in the denominator. Also, F(k) is entirely made of cosines, i.e. all the formally 
complex numbers are in fact simply real valued, due to the even symmetry of k about the 
origin. 

A simple and effective convolution approximation to Dk can be derived as follows. 
We convolve the left-hand and right-hand sides of Equation 2 by a kernel g. If a g exists 
such that k*g is a delta function at the origin (and zero everywhere else), then simply 
convolving an observed image A’ with g will restore the desired true image A.  If we 
substitute a delta function at the origin for A’ in Equation 2, then g will equal A and can 
be computed from Equation 4. The dimensions of g should be greater than or equal to 
those of k. For the usual IPC model, the kernel k is zero everywhere except for the center 
pixel and the four adjacent pixels that each share a side with the center pixel (Equation 
1). In the latter case, the kernel g is also very compact, with only a few elements 
significantly greater than zero.  For example, if k is parameterized by a coupling α and 
we select a 5x5 dimension for potentially non-zero values of g, then we can set up a set of 
6 linear equations and solve them for the six unique elements of g. We do this explicitly 
in the Appendix. To first order,  the approximate form of a 3x3 element g is 1+4α in the 
center element, –α in each of its four adjacent-side elements, and 2α2 in the corner 
elements adjacent to the center one. The kernel g should be normalized such that the sum 
of its elements is unity, so that convolution with it conserves counts while rearranging 
them within the array. 

An IDL implemetation of Equation 4 for 2-D images is given in the Appendix. The 
implementation is unlikely to be efficient in computational speed or memory allocation,  
and the bookkeeping of indices in the procedure is tedious. On a MacBook Pro purchased 
in 2007,  this algorithm requires 50 N ln N nanoseconds to deconvolve a square image of 
N pixels; e.g. a 1024x1024 pixel image (N=1 million) requires 0.6 seconds. The scaling 
is consistent with expectations for the FFT in the algorithm’s core. The convolution 
approximation (i.e. using kernel g) should be even faster and perhaps more robust against 
numerical artifacts related to round off errors in the many sums and products in the FFT. 

Figure 1 shows a simple, simulated “true” image, its simulated IPC-affected version 
similar to that which one would obtain from an IR array detector, and three versions of 
the latter image deconvolved with the IPC underestimated, known precisely, and 

overestimated. The purpose of this simulation is primarily to verify that the IDL 
implementation of the Fourier deconvolution is mathematically correct; the bookkeeping 
is the hardest part. 

 
 
 

 
Figure 1:  Simulations of IPC and deconvolution of IPC. Top row, left to right: a) the true image 
(“image A”) consists of a bad pixel (5000 ADU), bad pixels in a rectangle,  a column, and a slant (all 
10000 ADU), a row of bad pixels (200 ADU above background), and a background gradient (total 
change = 240 ADU), b) the image A convolved with a 3x3-pixel IPC kernel with alpha=0.02, c) image 
B superposed with Gaussian random noise (20 ADU rms). Bottom row, left to right: image C 
deconvolved with … d) alpha=0.016, showing positive residuals adjacent to bright pixels, e) 
alpha=0.020, showing optimum deconvolution, and f) alpha=0.024, showing negative residuals 
adjacent to bright pixels. The optimally deconvolved image E is very nearly equal to the true image A 
with the Gaussian random noise superposed, i.e. the effects of the IPC apparent in image B have been 
removed. The difference of image E and the true image with noise superposed (but with no IPC 
convolution) has mean, median, and rms values equal to -0.01, 0.05, and 2.0 ADU, respectively. An 
artifact of wrapping (i.e. periodic boundary conditions) in the convolution is evident in a few pixels in 
the upper-left and lower-right corners of images B, C, D, and F.  

Astronomers typically do not use Fourier deconvolution.Astronomical imagery tends 
to consist of isolated objects of interest (stars, galaxies) detected at low signal to noise 
ratio amongst many times as many pixels of blank sky plus noise. The zeroes in the 

Fourier transforms of telescope point spread functions and the stochastic noise (both 
detector noise and shot noise from the sky and the sources themselves) limit the 
usefulness of Fourier deconvolution of an optical point-spread-function such as the 
original, spherically-aberrated HST PSF. Importantly, because the IPC is a deterministic 
process, i.e. noiseless, and not a stochastic process, i.e. with inherent randomness or 
noise, IPC is corrected well with Fourier deconvolution, as demonstrated in the next 
section with authentic imagery. 

An Example from WFC3 IR, a H1RG Detector 
We obtained images with a Teledyne H1RG detctor, IR-1, during WFC3 thermal 

vacuum tests (“TV2”) at Goddard Space Flight Center (GSFC). Figure 2 illustrates a 
region of interest of one such image in its original state and after deconvolution according 
to Equation 4 with the IDL procedure in the Appendix.  The deconvolution used 
alpha=beta=0.01; no attempt was made to determine an optimum value of either alpha or 
beta. 

 
 

 
Figure 2:  A 336x72 pixel region of WFC3 IR-1. Top to bottom: the original image, the deconvolved 
image, and the difference between the two. Due to IPC, “hot” pixels appear as 5-pixel crosses, or + 
symbols, in the top image but as single-pixel peaks in the deconvolved (middle) image. The horizontal 
features, which look like three parallel “scratches,” have between 1000 and 3000 ADU per pixel, 

~100-times larger than the noise, ~10 ADU rms. At the right edge of the image is a nearly uniform 5-
pixel-wide band of reference pixels, which by design are insenstive to light.  

 
 

 
Figure 3:  Same as Figure 2, but with a spot of light from the CASTLE simulator of HST’s optical 
telescope assembly. The spot (14000 ADU peak) is slightly more extended in the original image (top) 
than the deconvolved image (middle). Because the deconvolution both concentrates and conserves 
flux, the difference image (bottom; original – deconvolved) is negative in the spot’s core (-60 ADU) 
which is surrounded by a positive ring (+30 ADU). The grey scale at the bottom refers to the 
difference image. 

Figure 3 displays a different region of the same WFC3 IR detector displayed in Figure 2, 
before and after the same deconvolution of IPC. A spot of light similar to a compact 
galaxy becomes slightly more concentrated after deconvolution, which effectively moves 
charge from the wings of the spot into its core, while conserving its flux. The 
concentration is not apparent in the images themselves but is apparent in their difference 
(cf. Figure 3). 

Linearity of Convolution and Fourier Transforms 
Linearity is useful for anticipating the potential effects of inserting the Fourier 

deconvolution within a data reduction pipeline. The convolution operator (*) and the 

Fourier transform F and its inverse F-1 are all linear in the mathematical sense. For 
example, F(c f + (g+h) * k) = c F(f) + F(g)F(k)+F(h)F(k), where f, g, and h are vectors or 
arrays (e.g. images), c is a scalar, and we have applied the convolution thereom of 
Fourier transforms, i.e. F(g*k) = F(g)F(k) as we did before in Equation 3. Let us revise 
Equation 2 to model the multiplication of a flat field AF,  the addition of a dark current 
image AD, followed by the convolution representing the IPC effect,  

(5) A’ = ( AF A + AD )*k. 
Recall that we measure A’ but we want to know A, which is proportional to the 
irradiance on the detector.  We also measure the dark current and the flat field, both of 
which are also affected by IPC and necessarily convolved with k, 

(6) AD’ = AD*k, and 
(7) AF’ = AF*k. 

It matters whether the flat-field image one uses is affected by IPC (denoted AF’), or has 
had the ffects of IPC removed by deconvolution (i.e. AF). The following equation, which 
represents (an approximation to) current practice, is incorrect: 

(8) A = (A’ -  AD’) / AF’. 
In the mathematical model of IPC represented by Equations 5-7, the correct dark 
subtraction and flat-field calibration is as follows: 

(9) A = Dk(A’ - AD’) / Dk(AF’),  
where the deconvolution operator Dk is defined in Equation 4. By linearity, additive 
terms, such as dark-current images, need not be deconvolved before subtraction from 
science images, although they can be, with identical results, i.e. in Equation 9, Dk(A’) - 
Dk(AD’) is identical to Dk(A’ - AD’). The deconvolution also is unaffected by any scalar 
gain correction. It is understood that the measured flat field will have its own (typically 
insignificant) dark current contribution, which we have neglected for simplicity. In the 
next section, we qualitatively outline potential implications of IPC and of using the 
correct Equation 9 instead of the incorrect Equation 8 for very precise astrometry and 
photometry. In a subsequent report, we will investigate those implications quantitatively. 

Discussion of Potential Benefits, Risks, and Unknowns of Deconvolution 
of IPC 

A case study illustrates a potential benefit of deconvolution of IPC. During the second 
thermal vacuum test (TV2) of WFC3, data from H1RG sensor IR-1 exhibited three 
parallel “scratches” (Figure 2).  At first inspection, the width of the top “scratch” appears 
to be resolved at left-of-center, unresolved at middle-right, and resolved again at far-
right. The deconvolved image gives a different impression: the “scratches” are 
everywhere unresolved in width. That the “scratches” are spatially resolved in the 
original image is an illusion caused by IPC and a hard stretch on the grey scale of the 
display. The pixels with ~3000 ADU produce ~30 ADUs of charge in each of the four 
adjacent pixels, which is 3-times the noise and thereby noticeable. The pixels in the 

middle-right section of the top “scratch” have only ~1000 ADU, and thereby produce a 
less-noticeable ~10 ADU of charge in their adjacent pixels, making that section of the 
“scratch” appear unresolved in the original image. The purpose of this case study is to 
illustrate how IPC can (and presumably may) lead to misinterpretation of data, even from 
astronomers who are aware of IPC: based upon the data prior to deconvolution, the 
author had thought the scratches were of various widths, indicating scratches of various 
depths into the detector surface. 

Cosmic rays create charge carriers, electrons and holes, as they pass through the 
detector material. The charge is localized both in time and in space. The nominal tactic to 
identify and subtract counts associated with cosmic-ray hits depends on sensing 
discontinuities in the rate of charge accumulation between successive samples of a 
MULTIACCUM sequence. Even if a cosmic ray passes through and creates charge in 
only a single pixel, five pixels will be affected because of IPC. Deconvolution of IPC 
should be able to restore the charge to the pixel(s) in which it was collected.3 In so far as 
that restoration is expected to be noiseless, the number of pixels expected to be flagged 
due to cosmic rays should be a few times less with deconvolution than without.  

Precision differential time-series photometry (e.g. of exoplanet transits) is already at 
or near the 100 ppm level with HST NICMOS NIC3. The technique generally involves 
spreading the incident light over many pixels to reduce the irradiance to manageable 
levels (to avoid saturation) and to smooth-over inaccuracies in the flat fields that affect 
differential photometry due to tiny instabilities in the pointing, i.e. undesired dithering of 
the image occurs, typically at sub-pixel amplitudes. Whereas it is advantageous for the 
incident irradiance to be uniform, the smoothing effect of IPC is expected to cause 
incorrect flat fielding of images from sensors with non-uniform pixel-to-pixel 
sensitivities if the data processing is based on Equation 8. An example of one such 
problem is the “sawtooth” effect in photon transfer curves described by McCullough et 
al. (2007). 

For a symmetric IPC kernel, k, astrometry should not be affected whether 
deconvolution is applied or not, except possibly at a very detailed level due to non-
linearities or biases in a centroiding algorithm’s handling of the sky within a star’s 
photometric aperture  and in “second-order” effects related to use of an incorrect flat field 
(vis. Equation 8). Without deconvolution, the second moments (and variances) of the 
observed stellar profiles will be the sum of the second moments (and variances) of the 
stars and the IPC kernel. For example, if the variance (in pixels2) of the charges’ spatial 
distribution in the true image of an object is (VX,VY), then the IPC-affected variance 
(VX’,VY’) = (VX,VY) + (2α,2β). Because noise is superposed by the readout process after 
the IPC convolution has occurred, deconvolution of that noise is incorrect. For an object 
observed with a large signal-to-noise ratio, deconvolution should restore the variance to 
                                                             
3 Due to charge diffusion, we cannot know with certainty where a charge was created but 
only where it was collected. 

the value it would have been with no IPC, but in low signal-to-noise ratio cases we 
expect to show (in future work) that the variance may be more accurately estimated by 
measuring (VX’,VY’) with the noisy data and then estimating (VX,VY) by subtracting 
(2α,2β), instead of measuring (VX,VY) directly from deconvolved images.  

Conclusions 
We propose a method for reducing or eliminating the effects of interpixel capacitance 

on infrared array data by deconvolution. Reduction of the number of pixels affected by 
each cosmic-ray hit, often from five pixels to one pixel, may be the greatest benefit of the 
deconvolution technique proposed in this work, or future extensions to it that operate on a 
cube of MULTIACCUM readouts. We predict that if flat-field images are deconvolved to 
remove the effects of IPC, photomery and astrometry may be improved slightly. We also 
anticipate that deconvolution of IPC from flat-fields images should result in images for 
which the constant of proportionality between the variance and the mean signal in each 
pixel is commensurate with the system gain, i.e. photon transfer curves and quantum 
efficiences derived from them will not need to be corrected for IPC because its effects 
will have been removed from the data by deconvolution.  Similarly, the values of pixels 
within extended sources should exhibit the expected Poisson noise, if IPC is 
deconvolved, but not otherwise. In general, all pixels are illuminated by an extended 
source, the diffuse zodiacal foreground. For pixels adjacent to ``hot’’ pixels,  
deconvolving IPC from dark-current images may facilitate distinguishing the 
deterministic IPC cross-talk of an adjacent ``hot’’ pixel from that of  a  ``warm’’ pixel. 
All the results and predictions of this report are predicated upon the model of IPC acting 
as a convolution. While that model has a good physical rationale and has been 
demonstrated approximately in prior work, additonal detailed studies are warranted to 
better characterize the effects of IPC and techniques for their removal such as the one 
described in this report. 

The author appreciates suggestions for improvement of the manuscript by Susana 
Deustua and Ron Allen. 

Appendices: IDL procedures (three) 

Procedure 1 
;+ 
; 
; Experiments with direct Fourier inversion of convolution to remove 
; IPC (interpixel capacitance) from images. 
; 
; To demonstrate it, type "ipcecon_demo" at an IDL prompt. 
; This also requires the usual Goddard IDL library for FITS files, etc. 
; 
; P. R. McCullough 2007 
;- 
 
pro ipcdecon,im,ipcpsf,alpha=alpha,beta=beta 
 
; as a convenience to the user, create the ipcpsf as a [3,3] matrix 
; from alpha and optionally also beta if different than alpha. 
if n_elements(alpha) gt 0 then begin 
  if n_elements(beta) eq 0 then beta = alpha 
  ipcdim = 3 
  ipcpsf = [0,beta,0,alpha,1.-2.*(beta+alpha),alpha,0,beta,0] 
  ipcpsf = reform(ipcpsf,ipcdim,ipcdim) 
endif 
 
si1 = size(im) 
si2 = size(ipcpsf) 
 
; bias the image to avoid negative pixel values in the image, which the 
; FFT method of deconvolution has trouble with. 
 
min_im = min(im) 
im     = im - min_im 
 
ipc_big = im*0. 
sx = si1[1]/2 - si2[1]/2 
sy = si1[2]/2 - si2[2]/2 
ipc_big[sx:sx+si2[1]-1,sy:sy+si2[2]-1] = ipcpsf 
 
ft_im  = fft(im) 
ft_psf = fft(ipc_big) 
im = shift(fft(ft_im/ft_psf,/inverse),-si1[1]/2,-si1[2]/2)/(float(si1[1])*float(si1[2])) 
 
; convert from Complex to Real 
im = float(im) 
 
; restore by removing the bias 
im = im + min_im 
 
return 
end 

Procedure 2 
;+ 
; 
; Experiments with direct Fourier inversion of convolution to remove 
; IPC (interpixel capacitance) from images. 
; 
; To execute it, simply type "ipcecon_demo" at an IDL prompt. 
; This requires ipcdecon.pro and the usual Goddard IDL library for FITS files, etc. 
; 
; P. R. McCullough 2007 
;- 
 
pro ipcdecon_demo,nx=nx,ny=ny,alpha=alpha,beta=beta,dalpha=dalpha,dbeta=dbeta,rms=rms 
 
if n_elements(nx) eq 0 then     nx = 32 
if n_elements(ny) eq 0 then     ny = 32 
if n_elements(alpha) eq 0 then alpha = 0.02 
if n_elements(beta) eq 0 then  beta = 0.02 
if n_elements(dalpha) eq 0 then dalpha = alpha*0.2 
if n_elements(dbeta) eq 0 then  dbeta = beta*0.2 
if n_elements(rms) eq 0 then    rms = 20.  ; electrons, rms readout noise 
 
im = fltarr(nx,ny) 
npix = n_elements(im) 
 
; scratch along a row, intentionally additive to the gradient (next) 
im[*,ny*3/4:ny*3/4] = 200. 
 
; zodiacal light gradient 
;gridmaker,nx,ny,1,1,ix,iy 
im = im + 12.*rms*findgen(npix)/float(npix) 
 
; bad pixel 
im[ny*3/4-5,ny*3/4-5] = 5000. 
 
; scratch along a column 
im[ny/8:ny/8,*] = 10000. 
 
; diagonal scratch 
im[indgen(ny)/8,indgen(ny)/3] = 10000. 
 
; rectangular "star" or "galaxy" 
im[nx/4+5:nx/4+5+2,ny/4+7:ny/4+7+3] = 10000. 
 
writefits,'ipcdecon_im.fits',im 
 
ipcdim = 3 
ipc = [0,beta,0,alpha,1.-2.*(beta+alpha),alpha,0,beta,0] 
ipc = reform(ipc,ipcdim,ipcdim) 
print,'Image convolved with IPC kernel ' 
print, ipc 
 
imc = convol(im,ipc,/edge_wrap) 
writefits,'ipcdecon_imc.fits',imc 

; add noise as if from a readout amplifier 
im_noise = rms*randomn(seed,n_elements(imc)) 
imc = imc + im_noise 
writefits,'ipcdecon_imcn.fits',imc 
 
for i = -1, 1 do begin 
 
  ii = strtrim(string(i),2) 
 
  ipc = [0,beta+i*dbeta,0,alpha+i*dalpha,1.-
2.*(beta+i*dbeta+alpha+i*dalpha),alpha+i*dalpha,0,beta+i*dbeta,0] 
  ipc = reform(ipc,ipcdim,ipcdim) 
 
  print,'ipcdecon_imd'+ii+'.fits was deconvolved with kernel ' 
  print, ipc 
  im_dec = imc 
  ipcdecon,im_dec,ipc 
  writefits,'ipcdecon_imd'+ii+'.fits',im_dec 
  im_dif = im + im_noise - im_dec 
  writefits,'ipcdecon_dif'+ii+'.fits',im_dif 
  print,'Mean, median, stdev of residual image imdecon_dif'+ii+'.fits:' 
  print,mean(im_dif),median(im_dif),stddev(im_dif) 
 
endfor 
 
return 
end 

Procedure 3 

This procedure solves the set of linear equations δ = M v and then populates 
the elements of the convolution kernel g with its unique elements contained 
in the vector v = M-1 δ. Here are three examples. 
IDL> print,ipcg(0.01,gdim=3)  
   0.00022634675    -0.010864644   0.00022634675 
    -0.010864644       1.0425532    -0.010864644 
   0.00022634675    -0.010864644   0.00022634675 
IDL> print,ipcg(0.01,gdim=5) 
   7.3705594e-08  -3.5378686e-06   0.00011321180  -3.5378686e-06   7.3705594e-08 
  -3.5378686e-06   0.00022634989    -0.010861257   0.00022634989  -3.5378686e-06 
   0.00011321180    -0.010861257       1.0421148    -0.010861257   0.00011321180 
  -3.5378686e-06   0.00022634989    -0.010861257   0.00022634989  -3.5378686e-06 
   7.3705594e-08  -3.5378686e-06   0.00011321180  -3.5378686e-06   7.3705594e-08 
IDL> print,ipcg(0.02,gdim=5) 
   1.4669064e-06  -3.3738849e-05   0.00051732903  -3.3738849e-05   1.4669064e-06 
  -3.3738849e-05    0.0010331912    -0.023729658    0.0010331912  -3.3738849e-05 
   0.00051732903    -0.023729658       1.0889806    -0.023729658   0.00051732903 
  -3.3738849e-05    0.0010331912    -0.023729658    0.0010331912  -3.3738849e-05 
   1.4669064e-06  -3.3738849e-05   0.00051732903  -3.3738849e-05   1.4669064e-06 

;+ 
; 
; Convolution with kernel g removes IPC (interpixel capacitance) from images. 
; 
; Solve for the 3x3 or 5x5 matrices of a kernel g for which 
; k*g = delta function at origin. 
; 
; P. R. McCullough 2008 
;- 
 
function ipcg,alpha,gdim=gdim 
 
if n_elements(gdim) eq 0 then gdim = 5 
 
; define shorthand variables 
  a  = alpha*1.d0 
  a2 = a*2.d0 
  a4 = a*4.d0 
  aa = 1.-a4 

 
if gdim eq 3 then begin 
 
  m = [[aa,a4,0.],[ a,aa,a2],[0.,a2,aa]] 
  v = transpose(invert(m))#[1.,0.,0.] 
  g = [[v[2],v[1],v[2]],[v[1],v[0],v[1]],[v[2],v[1],v[2]]] 
  g = g/total(g) 
 
endif 
 
if gdim eq 5 then begin 
 
  m = [[aa,a4,0.,0.,0.,0.], $ 
       [ a,aa,a2, a,0.,0.], $ 
       [0.,a2,aa,0.,a2,0.], $ 
       [0.,a ,0.,aa,a2,0.], $ 
       [0.,0.,a ,a ,aa,a ], $ 
       [0.,0.,0.,0.,a2,aa] ] 
 
  v = transpose(invert(m))#[1.,0.,0.,0.,0.,0.] 
  g = [[v[5],v[4],v[3],v[4],v[5]], $ 
       [v[4],v[2],v[1],v[2],v[4]], $ 
       [v[3],v[1],v[0],v[1],v[3]], $ 
       [v[4],v[2],v[1],v[2],v[4]], $ 
       [v[5],v[4],v[3],v[4],v[5]] ] 
  g = g/total(g) 
 
endif 
 
return,g 
end 
 


